

## 5. FUNCIONES DE VARIAS VARIABLES.

### EJERCICIOS

EJERCICIO 5.1. Estúdiase la existencia del límite en (0,0) de la función  $f: \mathbb{R}^2 \rightarrow \mathbb{R}$  definida como

$$f(x,y) = (y + y^2)/(x^2 - 3y).$$

SOLUCIÓN. Se define f

```
(%i1) define(f(x,y),(y+y^2)/(x^2-3*y));
```

```
(%o1) f(x,y) := 
$$\frac{y^2 + y}{x^2 - 3y}$$

```

Con Maxima no se pueden calcular límites de funciones de varias variables. Sustituyendo la variable x por 0 y la variable y por 0, resulta

```
(%i2) f(0,0);
```

```
expt: undefined: 0 to a negative exponent.
```

```
#0: f(x=0,y=0)
```

```
-- an error. To debug this try: debugmode(true);
```

y, por tanto, hay que estudiar la existencia del límite por otros métodos. Los límites reiterados de f en (0,0) son:

```
(%i3) limit(f(x,y),x,0); limit(%y,0);
```

```
limit(f(x,y),y,0); limit(%x,0);
```

```
(%o3) 
$$-\frac{y+1}{3}$$

```

```
(%o4) 
$$-\frac{1}{3}$$

```

```
(%o5) 0
```

```
(%o6) 0
```

Como los límites reiterados no coinciden, no existe el límite de f en (0,0).

EJERCICIO 5.2. Estúdiase la existencia del límite en (0,0) de la función  $f: \mathbb{R}^2 \rightarrow \mathbb{R}$  definida como

$$f(x,y) = y^2 x^2 / (x^6 + y^3).$$

SOLUCIÓN. Se define f:

```
(%i7) define(f(x,y), y^2*x^2/(x^6+y^3));
```

```
(%o7) f(x,y) := 
$$\frac{x^2 y^2}{y^3 + x^6}$$

```

En  $(0,0)$ , al sustituir resulta  $0/0$ , luego es una indeterminación. Los límites reiterados de  $f$  en  $(0,0)$  son:

```
(%i8) limit(f(x,y),x,0); limit(%y,0);
```

```
limit(f(x,y),y,0); limit(%x,0);
```

```
(%o8) 0
```

```
(%o9) 0
```

```
(%o10) 0
```

```
(%o11) 0
```

Como los límites reiterados coinciden con el valor 0, en caso de existir el límite de  $f$  cuando  $(x,y)$  tiende hacia  $(0,0)$ , éste es 0, pero falta demostrar la existencia del mismo. Calculando el límite de  $f$  en  $(0,0)$  a través de las rectas  $y=mx$ , con  $m$  número real, se tiene que

```
(%i12) limit(f(x,m*x),x,0);
```

```
(%o12) 0
```

De nuevo, se deduce que en caso de existir el límite, éste es 0. Si se calcula el límite de  $f$  en  $(0,0)$  a través de curvas de ecuación  $y=mx^2$  se obtiene

```
(%i13) limit(f(x,m*x^2),x,0);
```

```
(%o13)  $\frac{m^2}{m^3 + 1}$ 
```

y como este último límite depende del parámetro  $m$ , se concluye que no existe el límite de  $f$  en  $(0,0)$ .

EJERCICIO 5.3. Estúdiese la continuidad de la función  $f: \mathbb{R}^2 \rightarrow \mathbb{R}$  definida como

$$f(x,y) = x^2 y / (x^2 + y^2), \text{ si } (x,y) \text{ es distinto de } (0,0), \quad f(0,0) = 0.$$

SOLUCIÓN. La función  $f$  es continua en  $\mathbb{R}^2 \setminus \{(0,0)\}$  por ser cociente de polinomios y no anularse el denominador. Con respecto al punto  $(0,0)$ , la función  $f$  será continua en  $(0,0)$  si existe el límite de  $f$  en  $(0,0)$  y éste coincide con  $f(0,0)=0$ .

Se define  $f$  para los puntos  $(x,y)$  distintos de  $(0,0)$ :

```
(%i14) define(f(x,y),x^2*y/(x^2+y^2));
```

```
(%o14)  $f(x,y) := \frac{x^2 y}{y^2 + x^2}$ 
```

Los límites reiterados de  $f$  en  $(0,0)$  son

```
(%i15) limit(f(x,y),x,0); limit(%o,y,0);
```

```
limit(f(x,y),y,0); limit(%o,x,0);
```

```
(%o15) 0
```

```
(%o16) 0
```

```
(%o17) 0
```

```
(%o18) 0
```

Así pues, se deduce que en caso de existir el límite de  $f$  en  $(0,0)$ , éste es 0.

Por el modo en que está definida la función  $f$  (aparecen sumas de cuadrados de las variables  $x$  e  $y$ ) parece lógico pensar en hacer un cambio a coordenadas polares.

Se define  $F(r,t)=f(rcos(t), rsent(t))$ :

```
(%i19) assume(r>0); define(F(r,t), f(r*cos(t),r*sin(t)));
```

```
(%o19) [r > 0]
```

```
(%o20) F(r,t):= 
$$\frac{r^3 \cos(t)^2 \sin(t)}{r^2 \sin(t)^2 + r^2 \cos(t)^2}$$

```

Si existe una función  $h(r)$  que tiende hacia 0 cuando  $r$  tiende hacia 0 tal que  $|F(r,t)-0| \leq h(r)$ , entonces se demuestra que el límite de  $f$  cuando  $(x,y)$  tiende hacia  $(0,0)$  es 0, con lo que  $f$  es continua en  $(0,0)$ . Así pues, se calcula  $|F(r,t)-0|$  y se simplifica (el comando `trigsimp` sirve para simplificar expresiones trigonométricas):

```
(%i21) abs(F(r,t)-0); trigsimp(%);
```

```
(%o21) 
$$\frac{r^3 \cos(t)^2 |\sin(t)|}{r^2 \sin(t)^2 + r^2 \cos(t)^2}$$

```

```
(%o22) 
$$r \cos(t)^2 |\sin(t)|$$

```

Es claro que  $\cos(t)^2 |\sin(t)| \leq 1$ , por tanto, definiendo  $h(r)=r$ , se tiene que  $h(r)$  tiende hacia 0 cuando  $r$  tiende hacia 0 y  $|F(r,t)-0| \leq h(r)$ . Luego el límite de  $f$  cuando  $(x,y)$  tiende hacia  $(0,0)$  es 0 y  $f$  es continua en  $(0,0)$ .

EJERCICIO 5.4 Sea  $f: \mathbb{R}^2 \rightarrow \mathbb{R}$  la función definida como

$$f(x,y) = yx^2 / (x^2 - y^2).$$

Calcúlese  $D_{(1,-1)}f(1,2)$ , la derivada direccional de  $f$  en el punto  $(1,2)$  siguiendo la dirección del vector  $(1,-1)$ .

SOLUCIÓN. Se define la función  $f$

```
(%i23) define(f(x,y), y*x^2/(x^2-y^2));
```

```
(%o23) 
$$f(x,y) := \frac{x^2 y}{x^2 - y^2}$$

```

Según la definición, la derivada direccional de  $f$  en el punto  $(1,2)$  siguiendo la dirección del vector  $(1,-1)$  es

```
(%i24) limit((f(1+h*1,2+h*(-1))-f(1,2))/h, h, 0);
```

```
(%o24) -39/49
```

Obsérvese que la función  $f$  es diferenciable en  $(1,2)$ , pues es cociente de polinomios y en el punto  $(1,2)$  no se anula el denominador. Así pues, otra forma de determinar  $D_{(1,-1)}f(1,2)$  es a través de la igualdad  $D_{(1,-1)}f(1,2)=Gf(1,2)\cdot(1,-1)$ , donde  $Gf(1,2)$  denota el gradiente de  $f$  en el punto  $(1,2)$ .

Obtengamos también  $D_{(1,-1)}f(1,2)$  a través de la relación anterior. Se determina el gradiente de  $f$  en el punto  $(1,2)$ :

```
(%i25) define(Gf(x,y), jacobian([f(x,y)], [x,y])); Gf(1,2);
```

```
(%o25) Gf(x,y):=[ 2 x y / (x-2 y^2)^2 - x^2 y / (x-2 y^2)^2, x^2 / (x-2 y^2)^2 + 4 x^2 y^2 / (x-2 y^2)^2 ]
```

```
(%o26) [-30/49, 9/49]
```

Se calcula el producto escalar  $Gf(1,2)\cdot(1,-1)$ :

```
(%i27) %.[1,-1];
```

```
(%o27) -39/49
```

EJERCICIO 5.5. Determinése el plano tangente a la gráfica de la función

$$f(x,y)=3-x^2+4x-4y^2$$

en el punto  $(2,2)$  y represéntense gráficamente la función y el plano tangente.

SOLUCIÓN. Se define  $f$

```
(%i28) define(f(x,y), 3-x^2+4*x-4*y^2);
```

```
(%o28) f(x,y):=-4 y^2 - x^2 + 4 x + 3
```

La función  $f$  es diferenciable en  $\mathbb{R}^2$  por ser polinómica. Luego el plano tangente a  $f$  en el punto  $(2,2)$  tiene ecuación

$$T(x,y)=f(2,2)+Gf(2,2)\cdot(x-2,y-2),$$

donde  $Gf(2,2)$  denota el gradiente de  $f$  en el punto  $(2,2)$ .

Se calcula el gradiente de  $f$ :

```
(%i29) define(Gf(x,y), jacobian([f(x,y)], [x,y]));
```

```
(%o29) Gf(x,y):=[ 4 - 2 x, - 8 y ]
```


Por tanto, la ecuación del plano tangente  $(T(x,y))$  viene dada por

```
(%i30) define(T(x,y), f(2,2)+Gf(2,2).[x-2,y-2]);
```

```
(%o30) T(x,y) := -16 (y-2) - 9
```

En la siguiente figura se muestra la gráfica de  $f$  y el plano tangente calculado para  $x$  e  $y$  pertenecientes al intervalo  $[-10,10]$  (observe cómo se introduce la instrucción `wxplot3d` para dibujar dos funciones en una misma gráfica).

```
(%i31) wxplot3d([f(x,y), T(x,y), [x,-10,10], [y,-10,10]]);
```



```
(%o31)
```

EJERCICIO 5.6. Sean las funciones  $f: \mathbb{R}^2 \rightarrow \mathbb{R}$  y  $g: \mathbb{R}^2 \rightarrow \mathbb{R}^2$  definidas, respectivamente, como

$$f(x,y) = x^3 - xy^2 + y \ln(x^2 + 1)$$

$$g(u,v) = (u^2 - 2uv, u+v).$$

Determinése el gradiente de la función composición de  $g$  con  $f$  ( $f(g(x,y))$ ) en el punto  $(1,1)$  aplicando la regla de la cadena.

SOLUCIÓN. Se definen  $f$  y  $g$ :

```
(%i32) define(f(x,y), x^3-x*y^2+y*log(x^2+1));
```

```
define(g(u,v), [u^2-2*u*v, u+v]);
```

```
(%o32) f(x,y) := -x y^2 + log(x^2+1) y + x^3
```

```
(%o33) g(u,v) := [u^2-2 u v, v+u]
```

Puesto que las funciones  $f$  y  $g$  son diferenciables en  $\mathbb{R}$  y en  $\mathbb{R}^2$ , respectivamente, aplicando la regla de la cadena, la función  $f(g)$  es diferenciable en  $(1,1)$  y se tiene que

$$\text{Gf}(g)(1,1) = \text{Gf}(g(1,1)) \cdot \text{Jg}(1,1),$$

donde  $\text{Gf}(g)(1,1)$  denota el gradiente de  $f(g)$  en  $(1,1)$ ;  $\text{Gf}(g(1,1))$  denota el gradiente de  $f$  en el punto  $g(1,1)$  y  $\text{Jg}(1,1)$  es la matriz jacobiana de  $g$  en el punto  $(1,1)$ .

Se determina primero el gradiente de  $f$ :

```
(%i34) define(Gf(x,y), jacobian([f(x,y)], [x,y]));
```

```
(%o34) Gf(x,y) := [ -y^2 + (2*x*y)/(x^2+1) + 3*x^2, log(x^2+1) - 2*x*y ]
```

$g(1,1)$  vale

```
(%i35) g(1,1);
```

```
(%o35) [ -1, 2 ]
```

Luego  $\text{Gf}(g(1,1))$  es igual a

```
(%i36) Gf(g(1,1)[1], g(1,1)[2]);
```

```
(%o36) [ -3, log(2) + 4 ]
```

A continuación, se obtiene la matriz jacobiana de  $g$  en el punto  $(1,1)$ :

```
(%i37) define(Jg(u,v), jacobian(g(u,v), [u, v])); Jg(1,1);
```

```
(%o37) Jg(u,v) := [ 2*u - 2*v, -2*u ]
                  [ 1, 1 ]
```

```
(%o38) [ 0, -2 ]
        [ 1, 1 ]
```

Por tanto,  $\text{Gf}(g)(1,1)$  es igual a

```
(%i39) %o36.%;
```

```
(%o39) [ log(2) + 4, log(2) + 10 ]
```

NOTA: Obsérvese que se puede obtener el gradiente pedido hallando primero la composición y luego el gradiente de la compuesta con las siguientes instrucciones:

```

(%i40) h(u,v):=f(g(u,v)[1],g(u,v)[2]);

          define(Gh(u,v), jacobian([h(u,v)], [u, v])); Gh(1,1);
(%o40) h(u, v):= f((g(u, v))_1, (g(u, v))_2)
(%o41) Gh(u, v):=
[ log((u^2 - 2 u v)^2 + 1) +  $\frac{2 (2 u - 2 v) (v + u) (u^2 - 2 u v)}{(u^2 - 2 u v)^2 + 1}$  + 3 (2 u - 2 v) (u^2 - 2 u v)^2 - (2 u - 2 v) (v + u)^2 - 2 (v + u) (
(%o42) [ log(2) + 4  log(2) + 10 ]

```

Como se puede ver, la expresión del gradiente de  $f(g)$  es muy larga. Aplicando la regla de la cadena los cálculos son mucho más sencillos.

EJERCICIO 5.7. Obtenga y clasifique los extremos relativos de la función

$$f(x,y) = -x^3 + 4xy - 2y^2 + 1.$$

SOLUCIÓN. Se define la función  $f$ :

```

(%i43) define(f(x,y), -x^3+4*x*y-2*y^2+1);
(%o43) f(x, y):= - 2 y^2 + 4 x y - x^3 + 1

```

La función  $f$  es diferenciable en  $\mathbb{R}^2$  por ser polinómica, luego los extremos relativos de  $f$  son en particular puntos críticos. Para obtener los puntos críticos, se calculan las derivadas parciales y se iguala a 0:

```

(%i44) solve([diff(f(x,y),x,1), diff(f(x,y),y,1)], [x,y]);
(%o44) [[x =  $\frac{4}{3}$ , y =  $\frac{4}{3}$ ], [x = 0, y = 0]]

```

Así pues, se obtienen los puntos críticos  $P1=(4/3, 4/3)$  y  $P2=(0,0)$ . Para clasificar estos dos puntos críticos, se estudia la matriz hessiana de  $f$  en cada uno de ellos:

```

(%i45) define(Hf(x,y), hessian(f(x,y), [x,y]));
(%o45) Hf(x, y):=  $\begin{bmatrix} -6x & 4 \\ 4 & -4 \end{bmatrix}$ 

```

En el punto  $P1$ , se tiene que

```

(%i46) Hf(4/3,4/3)[1,1]; determinant(Hf(4/3,4/3));
(%o46) - 8
(%o47) 16

```


Por tanto, la matriz hessiana en este punto es definida negativa y  $P1$  es un máximo relativo.  
En el punto  $P2$  resulta

```
(%i48) Hf(0,0)[1,1]; determinant(Hf(0,0));
(%o48) 0
(%o49) -16
```

Como el determinante de  $Hf(0,0)$  es negativo, se deduce que  $P2$  es un punto de silla.

En la siguiente figura se muestra la función  $f$  para  $x$  e  $y$  pertenecientes al intervalo  $[-2, 3]$ .

```
(%i50) wxplot3d(f(x,y), [x,-2,3], [y,-2,3]);
```



(%o50)

Si se ejecuta la siguiente instrucción, aparece una ventana que permite girar la gráfica en la que se aprecia mejor que  $P1$  es un máximo relativo y  $P2$  es un punto de silla.

```
(%i51) plot3d(f(x,y), [x,-2,3], [y,-2,3]);
(%o51)
```